



THE DIFFERENTIALGEOMETRY SOFTWARE PROJECT

Calculate the "Kretschmann scalar" $R_{abcd}R^{abcd}$.

Synopsis

- On a pseudo-Riemannian manifold with metric g , the "Kretschmann scalar" is a quadratic scalar invariant of the Riemann R tensor of g , defined as $R_{abcd}R^{abcd}$, where indices are contracted using the metric.
- In this worksheet we show how to calculate the Kretschmann scalar from a metric.

Example 1

Load the DifferentialGeometry package and its Tensor sub-package.

```
[> with(DifferentialGeometry): with(Tensor):
```

Some optional aesthetic preferences:

```
[> interface(typesetting = extended):
```

```
[> Preferences("TensorDisplay", 1):
```

Define a coordinate chart on a 2-dimensional manifold.

```
[> DGsetup([theta, phi], S);
```

Manifold: S

(1

Define the metric. This is the standard "round" metric on the 2-sphere.

$$\begin{aligned} \text{S} > \mathbf{g} := \text{evalDG}(\text{dtheta} \& \text{t dtheta} + \sin(\text{theta})^2 \text{dphi} \& \text{t dphi}); \\ & \mathbf{g} := d\theta \otimes d\theta + \sin(\theta)^2 d\phi \otimes d\phi \end{aligned} \quad (2)$$

Calculate the Riemann curvature tensor of the metric g.

$$\begin{aligned} \text{S} > \mathbf{R} := \text{CurvatureTensor}(\mathbf{g}); \\ & \mathbf{R} := \sin(\theta)^2 \partial_\theta \otimes d\phi \otimes d\theta \otimes d\phi - \sin(\theta)^2 \partial_\theta \otimes d\phi \otimes d\phi \otimes d\theta - \partial_\phi \otimes d\theta \otimes d\theta \otimes d\phi + \partial_\phi \otimes d\theta \otimes d\phi \otimes d\theta \end{aligned} \quad (3)$$

Calculate the Kretschmann scalar, which is the metric inner product of the curvature with itself, $R_{abcd}R^{abcd}$. The TensorInnerProduct command takes the tensor product of any two tensors of the same type and then forms the scalar by contracting all indices using the metric.

$$\begin{aligned} \text{S} > \text{TensorInnerProduct}(\mathbf{g}, \mathbf{R}, \mathbf{R}); \\ & 4 \end{aligned} \quad (4)$$

Of course, one can also do all the tensor products and contractions "by hand":

$$\begin{aligned} \text{S} > \mathbf{h} := \text{InverseMetric}(\mathbf{g}); \\ & \mathbf{h} := \partial_\theta \otimes \partial_\theta + \frac{1}{\sin(\theta)^2} \partial_\phi \otimes \partial_\phi \end{aligned} \quad (5)$$

$$\begin{aligned} \text{S} > \text{ContractIndices}(\mathbf{g} \& \text{t h} \& \text{t h} \& \text{t h}, \mathbf{R} \& \text{t R}, [[1,1], [2,5], [3,2], [4,6], [5,3], [6,7], [7, 4], [8, 8]]); \\ & 4 \end{aligned} \quad (6)$$

Example 2

This same procedure will compute the scalar for any metric on any manifold. One simply chooses the appropriate coordinate chart and defines the appropriate metric. A famous example is the Kretschmann scalar of the Schwarzschild metric for a black hole spacetime.

$$\begin{aligned} > \text{DGsetup}([t, r, \text{theta}, \text{phi}], \mathbf{M}); \\ & \text{Manifold: } M \end{aligned} \quad (7)$$

M > f := 1 - 2*m/r;

$$f := 1 - \frac{2m}{r} \quad (8)$$

M > g := evalDG(- f * dt &t dt + 1/f * dr &t dr + r^2*(dtheta &t dtheta + sin(theta)^2*dphi &t dphi));

$$g := \frac{-r + 2m}{r} dt \otimes dt - \frac{r}{-r + 2m} dr \otimes dr + r^2 d\theta \otimes d\theta + r^2 \sin(\theta)^2 d\phi \otimes d\phi \quad (9)$$

M > R := CurvatureTensor(g):

M > TensorInnerProduct(g, R, R);

$$\frac{48 m^2}{r^6} \quad (10)$$

A complete set of scalar curvature invariants can be computed using the command [RiemannInvariants](#).

Commands Illustrated

- [DGsetup](#), [evalDG](#), [CurvatureTensor](#), [TensorInnerProduct](#)

Related Commands

- [RiemannInvariants](#)

References

Release notes

- Worksheet was executed using DG build "USU001-1460 9:25:48.66 Fri 01/03/2020"

Authors

Charles Torre
Department of Physics
Utah State University
December 2016

